

```
10 REM 'Paint' by Conrad Chin
20 ON ERROR GOTO 4290
30 ON BREAK GOSUB 3570:DEFINT a-z
40 DIM s!(72),c!(72),xr(4),yr(4),r$(4),pal(15),as1(100),as2(100)
50 CALL &BC02:PEN 1:PAPER 0
60 GOSUB 410
70 GOSUB 810
80 GOSUB 2190
90 SPEED KEY 20,8:GOSUB 3970:ok=-1
100 WHILE ok
110 GOSUB 2240
120 IF com THEN GOSUB 2340:k$="":GOTO 320
130 x1=x:y1=y
140 GOSUB 1320
```



```

150 ch=((x1-x)OR(y1-y))
160 IF ch AND rb THEN GOSUB 2580:GOSUB 2540
170 IF ch THEN x2=x:y2=y:x=x1:y=y1:GOSUB 2190:x=x2:y=y2:GOSUB 2190:GOSUB 1370:k$
   =""
180 IF k$=c$ THEN GOSUB 960
190 IF k$="g" THEN GOSUB 750
200 IF k$="r" THEN GOSUB 2010:k$=""
210 IF k$=CHR$(127) THEN GOSUB 2090:k$=""
220 IF rb THEN GOSUB 350:GOTO 320
230 GOSUB 2240
240 IF k$="p" THEN GOSUB 1130
250 IF k$>"0"AND k$<"5" THEN GOSUB 2190:GOSUB 1970:GOSUB 2190
260 GOSUB 1410
270 IF k$=CHR$(13) THEN SOUND 129,150,10,4:GOSUB 2290
280 IF k$=CHR$(12) THEN GOSUB 3610
290 IF k$=CHR$(19) THEN GOSUB 3800
300 IF k$=CHR$(16) THEN ok=0
310 com=NOT(INKEY(21))AND NOT(rb)
320 WEND
330 CLG 0:col=15:x=320:y=200:fx=x:fy=y
340 GOSUB 1370:GOSUB 810:GOTO 80
350 IF k$="a" THEN GOSUB 2660
360 IF k$=CHR$(13) THEN GOSUB 2690
370 IF k$>"0" AND k$<"5" THEN GOSUB 2190:GOSUB 2580:GOSUB 1970:GOSUB 2540:GOSUB
2190
380 IF k$="p" THEN c1=col:GOSUB 1130:GOSUB 2700
390 RETURN
400 REM      *** INITIALISE ***
410 MODE 0:CLG 0:ref$="1234":r$(1)="1":r$(2)="2":r$(3)="3":r$(4)="4"
420 GOSUB 660
430 POKE 46312,0:DEG:BORDER 0
440 KEY DEF 50,0:KEY DEF 38,0:KEY DEF 9,0:KEY DEF 68,0,221:KEY DEF 16,0,32,32,16
450 RESTORE 460:FOR i=49 TO 52:READ a,b:KEY DEF a,0,i:KEY DEF b,0:NEXT
460 DATA 13,64,14,65,5,57,20,56
470 DEF FNa(f)=INT(RND*2*f)-f:DEF FNacs(f!)= -ATN(f!/SQR(-f!*f!+1))+90
480 ORIGIN 0,16,0,640,400,16
490 u$=CHR$(240):d$=CHR$(241):l$=CHR$(242):r$=CHR$(243):c$="s"
500 x=320:y=200:col=15:wa=15:sz=50:r=1:z=7:fx=x:fy=y
510 RESTORE 520:FOR i=0 TO 7:READ a,b:WINDOW #i,a,b,25,25:NEXT
520 DATA 1,20,2,4,6,8,10,10,14,15,12,12,17,20,19,20
530 FOR i=0 TO 7:PEN #i,15:NEXT
540 i$=CHR$(15)+CHR$(4)+"X Y S G P R "
550 SYMBOL 241,7,56,64,128,64,56,7,0:SYMBOL 242,224,28,2,1,2,28,224,0
560 SYMBOL 243,24,36,66,66,129,129,129,0:SYMBOL 244,24,36,66,66,66,36,24,0
570 SYMBOL 245,0,0,3,12,48,192,0,0:SYMBOL 246,0,126,66,66,66,66,126,0:SYMBOL 247
,255,129,129,129,129,129,129,255
580 com$=CHR$(15)+CHR$(15)+" R A T F "+CHR$(243)+" "+CHR$(244)+" "+CHR$(245)+"
"+CHR$(241)+CHR$(242)+" "
590 spe$="2":gra$="N"
600 col$="15":sp=6:ab=0:rb=0:com=0:mo=0
610 GOSUB 700
620 ul$=CHR$(15)+CHR$(4)+"_":el$=CHR$(15)+CHR$(0)+"_"
630 FOR i=0 TO 15:pal$=pal$+CHR$(14)+CHR$(i)+CHR$(247):NEXT:pal$=CHR$(15)+CHR$(1
5)+pal$+CHR$(14)+CHR$(0)
640 RETURN
650 REM      *** SET UP PALETTE ***
660 RESTORE 670:FOR i=0 TO 15:READ c:INK i,c:pal(i)=c:NEXT
670 DATA 0,1,2,4,6,8,9,10,12,13,14,16,18,20,24,26
680 RETURN
690 REM      *** SET UP TABLES FOR sin , cos and arccos ***
700 DEG:FOR i=0 TO 72:s!(i)=SIN(i*5):c!(i)=COS(i*5):NEXT
710 FOR i=0 TO 99:as1(i)=FNacs(i/100):NEXT
720 FOR i=0 TO 99:as2(i)=FNacs(-i/100):NEXT
730 as1(100)=0:as2(100)=180:RETURN
740 REM *** CHANGE GRAPH. PRINT. MODE ***
750 SOUND 129,150,10,4:mo=mo+1:IF mo=4 THEN mo=0
760 gra$=MID$("NXAO",mo+1,1)
770 PRINT#5,gra$;
780 GOSUB 3970
790 RETURN
800 REM      *** PRINT INFO LINE ***
810 PRINT i$;
820 PRINT#1,RIGHT$(" "+STR$(x),3);
830 PRINT#2,RIGHT$(" "+STR$(y),3);
840 PRINT#3,spe$;
850 PRINT#4,col$;
860 PRINT#5,gra$;
870 PRINT#6,ref$;
880 RETURN
890 REM      *** DRAW PALETTE ***
900 PRINT
910 PRINT pal$;
920 PRINT#7,MID$(STR$(col),2)
930 LOCATE 18,1:PEN 4:PRINT"P";
940 RETURN
950 REM      *** CHANGE SPEED ***
960 SOUND 129,150,10,4:r=r+1:IF r>2 THEN r=0
970 spe$=MID$("123",r+1,1):PRINT#3,spe$;
980 sp=r*4+2
990 RETURN
1000 REM *** CHANGE INK OF SELECTED PEN ***
1010 SOUND 129,150,10,4:c=pal(col):LOCATE 18,1:PEN#6,4:PRINT#6,CHR$(13)+" I";
1020 PEN 0:LOCATE x1,1:PRINT CHR$(247);
1030 k$="":WHILE k$<>CHR$(13)
1040 PRINT#7,RIGHT$("0"+MID$(STR$(c),2),2);
1050 k$=INKEY$
1060 c=c+(k$=d$)-(k$=u$):IF c>26 THEN c=0
1070 IF c<0 THEN c=26
1080 INK col,c
1090 WEND:k$="":pal(col)=c
1100 LOCATE 18,1:PRINT#6,CHR$(13)+" P";:PEN#6,15:PRINT#7,RIGHT$("0"+MID$(STR$(co
l),2),2);
1110 RETURN
1120 REM      *** SELECT NEW PEN ***
1130 x1=col+1:x2=x1:k$=" ":SPEED KEY 30,5:GOSUB 900:pe=15:inv=15
1140 PRINT CHR$(22)+CHR$(1);
1150 PRINT#7,RIGHT$("0"+MID$(STR$(col),2),2);
1160 WHILE k$<>CHR$(13)
1170 inv=inv XOR 15:pe=pe XOR inv:LOCATE x1,1:PEN pe:PRINT CHR$(247);

```



```

1180 k$=INKEY$:x2=x1+(k$=1$)-(k$=r$):IF x2=0 THEN x2=16
1190 IF x2=17 THEN x2=1
1200 IF k$="c" THEN GOSUB 1010
1210 IF k$="r" THEN PRINT CHR$(7);:GOSUB 660
1220 col=x1-1
1230 IF x2<>x1 THEN PEN 15:LOCATE x1,1:PRINT CHR$(247);:PRINT#7,RIGHT$("0"+MID$(
STR$(x2-1),2),2);
1240 x1=x2:WEND
1250 col=x1-1:PRINT CHR$(22)+CHR$(0)
1260 SPEED KEY 20,8
1270 col$=RIGHT$("0"+MID$(STR$(col),2),2)
1280 IF ab=0 THEN PRINT CHR$(23)+CHR$(1);:PLOT fx,fy,wa:PLOT fx,fy,col
1290 GOSUB 3970
1300 wa=col:k$="":GOSUB 810:PRINT CHR$(23)+CHR$(mo):RETURN
1310 REM *** UPDATE X and Y COORDS ***
1320 k$=INKEY$
1330 x=x+((k$=r$)*sp*2*(x<(640-2*sp)))-((k$=l$)*sp*2*(x>2*sp-4))
1340 y=y+((k$=u$)*sp*(y<(384-sp)))-((k$=d$)*sp*(y>sp-2))
1350 RETURN
1360 REM *** PRINT X and Y COORDS ***
1370 PRINT#1,RIGHT$(" "+STR$(x),3);
1380 PRINT#2,RIGHT$(" "+STR$(y),3);
1390 RETURN
1400 REM *** AIRBRUSH ROUTINE ***
1410 IF NOT(ab) THEN RETURN ELSE GOSUB 2190:WHILE ab
1420 GOSUB 1520
1430 k$="":WHILE k$<>CHR$(13) AND ab
1440 FOR i=1 TO 5
1450 PLOT x+FNa(sz),y+FNa(sz),col:NEXT
1460 SOUND 129,0,30,3,0,0,1
1470 GOSUB 1320:GOSUB 1370
1480 WEND
1490 SOUND 129,0,0,0,0:GOSUB 1370
1500 WEND:SPEED KEY 20,8:GOSUB 2190
1510 GOSUB 3970:RETURN
1520 k$="":SPEED KEY 10,2
1530 GOSUB 1680:WHILE k$<>CHR$(13) AND ab
1540 x1=x:y1=y
1550 GOSUB 1320
1560 sz1=sz:sz=sz+((k$=CHR$(&F7))*2*(sz<150))-((k$=CHR$(&F6))*2*(sz>0))
1570 IF x1<>x OR y1<>y OR sz1<>sz THEN x2=x:y2=y:x=x1:y=y1:sz2=sz:sz=sz1:GOSUB 1
680:x=x2:y=y2:sz=sz2:GOSUB 1680:GOSUB 1370
1580 IF k$="p" THEN GOSUB 1130
1590 IF k$="g" THEN GOSUB 750
1600 IF k$="a" THEN ab=0
1610 IF k$>"0" AND k$<"5" THEN GOSUB 1680:GOSUB 1970:GOSUB 1680
1620 IF k$=c$ THEN GOSUB 960
1630 IF k$=CHR$(127) THEN GOSUB 2090:k$=""
1640 IF k$="f" THEN GOSUB 1730
1650 IF k$="r" THEN GOSUB 2010:k$=""
1660 WEND:GOSUB 1680
1670 SPEED KEY 1,1:RETURN
1680 PRINT CHR$(23)+CHR$(1);
1690 MOVE x-sz,y-sz:DRAW 0,2*sz,15:DRAW 2*sz,0:DRAW 0,-2*sz:DRAW -2*sz,0
1700 PRINT CHR$(23)+CHR$(mo);
1710 RETURN
1720 REM *** FILL IN SQUARE CURSOR ***
1730 GOSUB 1680:SOUND 129,150,10,4:FOR i=(y-sz) TO (y+sz) STEP 2
1740 MOVE (x-sz),i:DRAW 2*sz,0,col
1750 NEXT:GOSUB 1680
1760 GOSUB 3970
1770 RETURN
1780 REM *** DRAW FILLED ELLIPSE ***
1790 IF o2=0 THEN aa=0-180*(o1<0):GOTO 1840
1800 IF o1=0 THEN aa=90-180*(o2<0):GOTO 1840
1810 aa=ATN(o2/o1)
1820 IF aa<0 THEN aa=aa+180-180*(o2<0):GOTO 1840
1830 IF aa>0 THEN aa=aa-180*(o1<0)
1840 k=SQR(o1*o1+o2*o2)
1850 y1=0:bb=90+aa
1860 WHILE (y1/k)<1:bb=as1(100*y1/k)+aa
1870 x1=n1*COS(bb)-n2*SIN(bb)+fx:bb=-as1(100*y1/k)+aa:x2=fx+n1*COS(bb)-n2*SIN(bb)
1880 MOVE x1,y1+fy:DRAW x2,y1+fy,col
1890 y1=y1+2:WEND
1900 y1=-2:bb=90+aa
1910 WHILE (y1/k)>-1:bb=as2(-100*y1/k)+aa
1920 x1=n1*COS(bb)-n2*SIN(bb)+fx:bb=-as2(-100*y1/k)+aa:x2=n1*COS(bb)-n2*SIN(bb)+
fx
1930 MOVE x1,y1+fy:DRAW x2,y1+fy,col
1940 y1=y1-2:WEND
1950 RETURN
1960 REM *** CALL UP REFERENCE POINT ***
1970 k=ASC(k$)-48:IF r$(k)<>CHR$(246) THEN RETURN
1980 x=xr(k):y=yr(k)
1990 GOSUB 3970:RETURN
2000 REM *** SET UP REFERENCE POINT ***
2010 SOUND 129,150,10,4:WHILE (k$<"1" OR k$>"4") AND k$<>"a":k$=LOWER$(INKEY$):W
END
2020 IF k$="a" THEN 2070
2030 k=ASC(k$)-48:IF r$(k)=CHR$(246) THEN RETURN
2040 SOUND 129,150,10,4:xr(k)=x:yr(k)=y:r$(k)=CHR$(246)
2050 ref$=r$(1)+r$(2)+r$(3)+r$(4)
2060 PRINT#6,ref$;
2070 GOSUB 3970:RETURN
2080 REM *** DELETE REFERENCE POINT ***
2090 rr=0:FOR i=1 TO 4:IF r$(i)=CHR$(246) THEN rr=-1
2100 NEXT:IF NOT(rr) THEN 2170
2110 PRINT CHR$(7);:WHILE (k$<"1" OR k$>"4") AND k$<>"a":k$=LOWER$(INKEY$):WEND
2120 IF k$="a" THEN 2170
2130 k=ASC(k$)-48:IF r$(k)<>CHR$(246) THEN RETURN
2140 PRINT CHR$(7);:r$(k)=MID$(STR$(k),2)
2150 ref$=r$(1)+r$(2)+r$(3)+r$(4)
2160 PRINT#6,ref$;
2170 GOSUB 3970:RETURN
2180 REM *** PLOT CURSOR AND F.P. ***
2190 PRINT CHR$(23)+CHR$(1);
2200 MOVE x-8,y:DRAW 16,0,15:MOVER -8,-8:DRAW 0,16
2210 PLOT fx,fy,col
2220 PRINT CHR$(23)+CHR$(mo);

```



```

2230 RETURN
2240 PRINT CHR$(23)+CHR$(1);
2250 PLOT fx,fy,col
2260 PRINT CHR$(23)+CHR$(mo);
2270 RETURN
2280 REM *** SET POSITION OF F.P. ***
2290 GOSUB 2190:PRINT CHR$(23)+CHR$(1);
2300 fx=x:fy=y
2310 GOSUB 2190
2320 RETURN
2330 REM *** SELECT COMMAND ***
2340 LOCATE 1,1:sh=-1:z=15:PRINT com$;
2350 PRINT CHR$(22)+CHR$(1);:SPEED KEY 20,10
2360 GOSUB 2240
2370 WHILE sh AND k$<>CHR$(13)
2380 LOCATE z,1:PRINT el$;
2390 k$=INKEY$
2400 z=z+2*(k$=CHR$(&F6) AND z>3)-2*(k$=CHR$(&F7) AND z<17)
2410 LOCATE z,1:PRINT ul$;
2420 sh=NOT(INKEY(21))
2430 WEND
2440 PRINT CHR$(22)+CHR$(0);
2450 GOSUB 2470:com=0:GOSUB 3970
2460 SPEED KEY 10,1:RETURN
2470 IF k$<>CHR$(13) THEN GOSUB 810:RETURN
2480 i=(z-1)/2:ON i GOSUB 2630,2510,3210,4000,3360,2770,2750,2890
2490 GOSUB 810:RETURN
2500 REM *** SWITCH ON AIRBRUSH ***
2510 ab=-1
2520 RETURN
2530 REM *** ROUTINES FOR DRAWING RUBBER BAND ***
2540 PRINT CHR$(23)+CHR$(1);
2550 MOVE x,y:DRAW fx,fy,col:xa=x:ya=y
2560 PRINT CHR$(23)+CHR$(mo);
2570 RETURN
2580 PRINT CHR$(23)+CHR$(1);
2590 MOVE xa,ya:DRAW fx,fy,col
2600 PRINT CHR$(23)+CHR$(mo);
2610 RETURN
2620 REM *** SWITCH ON RUBBER BAND ***
2630 rb=-1:GOSUB 2190:GOSUB 2540:GOSUB 2190
2640 RETURN
2650 REM *** SWITCH OFF RUBBER BAND ***
2660 rb=0:GOSUB 2190:GOSUB 2580:GOSUB 2190
2670 GOSUB 3970:RETURN
2680 REM *** FIX RUBBER BAND ***
2690 rb=0:GOSUB 2190:GOSUB 2580:MOVE xa,ya:DRAW fx,fy,col:GOSUB 2190:RETURN
2700 GOSUB 2190:MOVE x,y:PRINT CHR$(23)+CHR$(1);
2710 DRAW fx,fy,col:DRAW x,y,col
2720 GOSUB 2190
2730 RETURN
2740 REM *** LINE ROUTINE ***
2750 GOSUB 2190:MOVE fx,fy:DRAW x,y,col:GOSUB 2190:GOSUB 2290:RETURN
2760 REM *** CIRCLE ROUTINE ***
2770 PRINT"SOLID (Y/N)?":WHILE k$<>"Y" AND k$<>"N" AND k$<>"A":k$=UPPER$(INKEY$):WEND
2780 IF k$="A" THEN RETURN
2790 solid=0:IF k$="Y" THEN solid=-1
2800 GOSUB 2190:dx=fx-x:dy=fy-y
2810 r1=SQR(dx*dx+dy*dy):MOVE fx+r1,fy:IF solid THEN 2840
2820 FOR i=1 TO 72:DRAW fx+r1*c!(i),fy+r1*s!(i),col:NEXT
2830 DRAW fx+r1,fy:GOTO 2870
2840 st=-r1:IF fy+st<0 THEN st=-fy
2850 fin=r1:IF fy+fin>383 THEN fin=383-fy
2860 FOR i=st TO fin STEP 2:x1=SQR(r1*r1-i*i):MOVE fx-x1,fy+i:DRAW fx+x1,fy+i,col:NEXT
2870 GOSUB 2190:RETURN
2880 REM *** ELLIPSE ROUTINE ***
2890 PRINT:PEN 15:INPUT;"Ratio";ec$:solid=0
2900 ec$=UPPER$(ec$):k=INSTR(ec$,"S"):IF k>0 THEN ec$=MID$(ec$,2):solid=-1:GOTO 2900
2910 f3=0:ec=VAL(ec$):IF ec<0 THEN ec=ABS(ec):IF NOT(solid) THEN f3=-1
2920 IF ec=0 THEN RETURN
2930 IF ec>999 OR ec<1 THEN RETURN
2940 IF f3 THEN GOSUB 3170 ELSE:sa=0:ea=72
2950 GOSUB 2190
2960 dx=x-fx:dy=y-fy:r1=SQR(dx*dx+dy*dy)
2970 IF dx=0 AND dy=0 THEN GOSUB 2190:RETURN
2980 r2=(ec/100)*r1
2990 IF dx=0 AND dy>0 THEN an=90:GOTO 3060
3000 IF dx=0 AND dy<0 THEN an=270:GOTO 3060
3010 IF dy=0 AND dx>0 THEN an=0:GOTO 3060
3020 IF dy=0 AND dx<0 THEN an=180:GOTO 3060
3030 an=ATN(dy/dx):IF dx<0 AND dy>0 THEN an=180+an:GOTO 3060
3040 IF dx<0 AND dy<0 THEN an=an+180:GOTO 3060
3050 IF dx>0 AND dy<0 THEN an=360+an
3060 n1=r1*COS(an):n2=r2*SIN(an)
3070 o1=r1*SIN(an):o2=r2*COS(an)
3080 IF solid THEN GOSUB 1790:GOTO 3150
3090 x0=n1*c!(sa)-n2*s!(sa)+fx:y0=o1*c!(sa)+o2*s!(sa)+fy
3100 MOVE x0,y0
3110 IF sa>ea THEN ea=ea+72
3120 FOR i=sa TO ea
3130 xn=n1*c!(i MOD 72)-n2*s!(i MOD 72)+fx:yn=o1*c!(i MOD 72)+o2*s!(i MOD 72)+fy
3140 DRAW xn,yn,col:NEXT
3150 GOSUB 2190
3160 RETURN
3170 PRINT:INPUT;"Start ";sa:IF sa>359 OR sa<0 THEN 3170
3180 PRINT:INPUT;"Finish ";ea:dif=(sa-ea):IF ea>359 OR ea<0 OR ABS(dif)<10 OR dif>350 THEN 3180
3190 ea=ea/5:sa=sa/5:RETURN
3200 REM *** PRINT TEXT ROUTINE ***
3210 GOSUB 810:GOSUB 2190:TAG#7
3220 y1=y-16:x1=x:SPEED KEY 30,2:GOSUB 3330
3230 k$="":WHILE k$<>CHR$(13)
3240 k$=INKEY$:IF k$="" THEN 3290
3250 GOSUB 3330:PLOT 700,0,col
3260 k=ASC(k$):IF k>31 AND k<127 AND x1<609 THEN MOVE x1,y:PRINT#7,k$;:x1=x1+32
3270 IF k=127 AND x1>31 THEN x1=x1-32:MOVE x1,y:PRINT#7," ";
3280 GOSUB 3330
3290 WEND
3300 TAGOFF#7:GOSUB 3330:GOSUB 2190

```



```

3310 SPEED KEY 20,8:RETURN
3320 REM *** DRAW TEXT CURSOR ***
3330 PRINT CHR$(23)+CHR$(1);:MOVE x1,y1:DRAW x1+32,y1,15:PRINT CHR$(23)+CHR$(mo)
;
3340 RETURN
3350 REM *** DRAW ARC ROUTINE ***
3360 PRINT:PEN 15:INPUT;"Angle ";an
3370 IF an<0 THEN an=ABS(an):f3=-1 ELSE f3=1
3380 IF an<5 OR an>355 THEN RETURN
3390 an=an/2:dx=(x-fx)*f3:dy=(y-fy)*f3:IF dx=0 AND dy=0 THEN RETURN
3400 GOSUB 2190
3410 a2=(SQR(dx*dx+dy*dy))/2
3420 IF an=90 THEN a3=0 ELSE a3=a2/TAN(an)
3430 IF dx=0 THEN a1=90-180*(dy<0):GOTO 3480
3440 IF dy=0 THEN a1=-180*(dx<0):GOTO 3480
3450 a1=ATN(dy/dx)
3460 IF dy<0 THEN a1=a1+180-180*(dx>0)
3470 IF dy>0 THEN a1=a1-180*(dx<0)
3480 r1=SQR(a3*a3+a2*a2)
3490 sa=90+a1-an:ea=90+a1+an
3500 IF f3=1 THEN a=x:b=y:a0=fx:b0=fy ELSE a=fx:b=fy:a0=x:b0=y
3510 x0=a-a2*COS(a1)+a3*SIN(a1)
3520 y0=b-a2*SIN(a1)-a3*COS(a1)
3530 MOVE a,b
3540 FOR i=sa TO ea STEP 5:DRAW x0+r1*COS(i),y0+r1*SIN(i),col:NEXT:DRAW a0,b0,col
1
3550 GOSUB 2190
3560 RETURN
3570 MODE 1:CALL &BC02:PEN 1:PAPER 0:CLS
3580 WINDOW 1,40,1,25
3590 KEY DEF 16,1,16,16,16:KEY DEF 9,1:CLS:SPEED KEY 35,2:PRINT " ":END
3600 REM *** LOAD IN NEW PICTURE ***
3610 PRINT:PRINT"<L>-LOAD , <A>-ABORT";
3620 WHILE INKEY$<>"":WEND
3630 k$="":WHILE k$<>"L" AND k$<>"A":k$=UPPER$(INKEY$):WEND
3640 IF k$="A" THEN 3780
3650 PRINT:GOSUB 2190
3660 LOAD"!
3670 OPENIN"!pal"
3680 FOR i=0 TO 15
3690 INPUT#9,pal(i)
3700 NEXT
3710 FOR i=1 TO 4
3720 INPUT#9,xr(i),yr(i),r$(i)
3730 NEXT
3740 CLOSEIN
3750 FOR i=0 TO 15:INK i,pal(i):NEXT
3760 ref$=r$(1)+r$(2)+r$(3)+r$(4)
3770 GOSUB 2190
3780 GOSUB 3970:GOSUB 810:RETURN
3790 REM *** SAVE CURRENT PICTURE ***
3800 PRINT:PRINT"<S>-SAVE , <A>-ABORT";
3810 WHILE INKEY$<>"":WEND
3820 k$="":WHILE k$<>"S" AND k$<>"A":k$=UPPER$(INKEY$):WEND
3830 IF k$="A" THEN 3960
3840 PRINT:SPEED WRITE 1:GOSUB 2190
3850 PRINT:PRINT " ** SAVING **";
3860 SAVE"!screen",b,49152,16384
3870 OPENOUT"!pal"
3880 FOR i=0 TO 15
3890 PRINT#9,pal(i)
3900 NEXT
3910 FOR i=1 TO 4
3920 PRINT#9,xr(i),yr(i),r$(i)
3930 NEXT
3940 CLOSEOUT
3950 GOSUB 2190
3960 GOSUB 3970:GOSUB 810
3970 WHILE INKEY$<>"":WEND:k$=""
3980 RETURN
3990 REM *** INFILL ROUTINE ***
4000 DIM x1(15),x2(15),y(15),d(15)
4010 GOSUB 2190
4020 sx=x:sy=y
4030 bg=TEST(x,y):d=2
4040 WHILE TESTR(-4,0)=bg AND XPOS>=0:WEND:x1=XPOS
4050 MOVE x,y:WHILE TESTR(4,0)=bg AND XPOS<640:WEND:x2=XPOS
4060 s=1:x1(1)=x1:x2(1)=x2:y(1)=y-d:d(1)=-d
4070 MOVE x1,y:k$=INKEY$:IF k$="a" THEN s=0:GOTO 4150
4080 WHILE TESTR(4,0)<>bg AND XPOS<x2:WEND:x3=XPOS-4:IF x3=x1 THEN MOVE x1+4,y:W
HILE TESTR(-4,0)=bg AND XPOS>=0:WEND:x3=XPOS
4090 IF (x1-x3)>4 THEN GOSUB 4180
4100 x1=x3:IF (x2-x1)<5 THEN 4150
4110 x3=x1
4120 MOVE x3,y:WHILE TESTR(4,0)=bg AND XPOS<640:WEND:x3=XPOS:MOVE x1+4,y:DRAW x3
-4,y,col
4130 IF ABS(x3-x2)>4 THEN GOSUB 4230
4140 x2=x3:y=y+d:IF y<0 OR y>383 THEN 4150 ELSE 4070
4150 IF s=0 THEN ERASE x1,x2,y,d:x=sx:y=sy:GOSUB 2190:RETURN
4160 x1=x1(s):x2=x2(s):y=y(s):d=d(s):s=s-1:GOTO 4070
4170 RETURN
4180 y1=y-d:MOVE x1,y1
4190 WHILE TESTR(-4,0)<>bg AND XPOS>x3:WEND:x4=XPOS
4200 IF x4=x3 THEN RETURN
4210 IF s<15 THEN s=s+1:x1(s)=x3:x2(s)=x4+4:y(s)=y1:d(s)=-d
4220 RETURN
4230 IF SGN(x3-x2)=1 THEN x4=x2:x5=x3:y1=y-d ELSE x4=x3:x5=x2:y1=y
4240 MOVE x4-4,y1:WHILE TESTR(4,0)<>bg AND XPOS<x5:WEND:x4=XPOS
4250 IF x4=x5 THEN RETURN
4260 IF s<15 THEN s=s+1:x1(s)=x4-4:x2(s)=x5:y(s)=y1:d(s)=SGN(x2-x3)*d
4270 RETURN
4280 REM *** ERROR HANDLING ROUTINE ***
4290 IF ERR=7 THEN mem=FRE(""):PRINT CHR$(7);:RESUME
4300 IF ERR=25 THEN PRINT"INCORRECT FILE"CHR$(7);:FOR i=1 TO 2000:NEXT:RESUME 37
70
4310 MODE 1:CALL &BC02:PEN 1:PAPER 0:CLS
4320 WINDOW 1,40,1,25
4330 KEY DEF 16,1,16,16,16:KEY DEF 9,1:CLS:SPEED KEY 30,2
4340 PRINT"ERROR NUMBER "ERR" IN LINE "ERL
4350 END

```